
     runCG7=()->(
	  	
     R = QQ[a,b,c,e,f,m,r,x,y,z,t];
     
     --Coefficients for myquintic (c.f. Catanese et. all)
     A = r^2;
     B = -(1/7)*(2*r^2-13*r-18);
     C = (73*r^2+75*r+92)*(1/49);
     E = -(r^2-24*r-9)*(1/7);
     F = (181*r^2+241*r+163)*(1/49);
     M = (3*r^2+5*r+1)*(1/7);
     
     A = sub(A,R);
     B = sub(B,R);
     C = sub(C,R);
     E = sub(E,R);
     F = sub(F,R);
     M = sub(M,R);
     
     quintic = (x+m*y+a*z)^2*t^3+(a^2*x^3+x*y*(b*x+c*y)+m^2*y^3+(e*x^2+f*x*y+c*y^2)*z+(b*x+e*y)*z^2+z^3)*t^2+(2*a*x^3*y+e*x^2*y^2+2*a*m*x*y^3+(2*a*m*x^3+f*x^2*y+f*x*y^2+2*m*y^3)*z+(c*x^2+f*x*y+b*y^2)*z^2+2*(m*x+a*y)*z^3)*t+x^3*y^2+a^2*x^2*y^3+x*y*z*(2*m*x^2+b*x*y+2*a*y^2)+z^2*(m^2*x^3+c*x^2*y+e*x*y^2+y^3)+(m*x+a*y)^2*z^3;

     myquintic = sub(quintic,{a=>A,b=>B,c=>C,e=>E,f=>F,m=>M});
     
     myquintic = myquintic*49; -- This makes myquintic a polynomial with only integer coefficients.

     myquintic = sub(myquintic,r=>143); -- [(143)^3 + (143)^2 -1](mod 7^3) == 0]
     
     R1 = ZZ/7[r,x,y,z,t];

     quintic2 = sub(myquintic,R1); -- Reduce myquintic mod 7.
     
     quintic3 = sub(quintic2,R); -- myquintic after reduction mod 7 back in QQ.
     
     taylor = myquintic-quintic3; -- First-order taylor over QQ.
     
     taylor = taylor/7; -- Divide by 7.
     
     taylor2 = sub(taylor,R1); -- Reduce first-order taylor mod 7.

     taylor3 = sub(taylor2, R); -- first-order taylor after reduction mod 7 back in QQ.
     
     taylor4 = myquintic-quintic3-7*taylor3; -- Second-order taylor over QQ.
     
     taylor4 = taylor4/49; -- Divide by 7^2.

     taylor5 = sub(taylor4,R1); -- Reduce second-order taylor mod 7.
     
     q = (x*z+y*t); -- Our favorite quadric.
     
     l = (x+y+z+t); -- Our favorite linear term.

     f3 = taylor2/q;

     f3 =  sub(f3,R1);

     f5 = taylor5;
     
     g = l*(q^2) + 7*q*f3 + (7^2)*f5; -- Our favorite 7-adic expression.
     
     disc = (f3)^2 - 4*l*f5; -- Discriminant of g, should be of degree 6.
     
     N = ideal(sub(disc,{t=>1,y=>-x*z})); -- Intersection of discriminant with quadric.
     
     N2 = sub(ideal(disc, q), t=>1);
     
     Sn = ideal(singularLocus(N));
     
     J = ideal(disc,q); -- Intersection of 7-adic discriminant with quadric.
      
     Nd = decompose(N); -- Decomposition of the ideal N.
     
     Jd = decompose(J); -- Decomposition of the idea J.
     
)	
